
Computer algebra and particle physics

• Developments in particle physics have gone hand in hand with
those in computer algebra

• In 1963 Veltman created SCHOONSCHIP
• It computed ``a monstrous expression involving in the order of
50 000 terms in intermediate stages'' and had to be stored on
tape

• In 1984 Vermaseren started work on FORM
• As computing power and algorithms improved, so did the
ambition for precision

• In the 1960s an order of magnitude agreement with experiment
was good

• Nowadays, the goal is to achieve < 1% error

Computational blow-up

• One of 6000 diagrams of the five-loop gluon propagator:

• Appyling Feynman rules yields 12 029 521 terms!
• Can be reduced to 23 master integrals with algebraic identities
• This reduction requires a terabyte of disk space and is
time-consuming

• Blow-up of rational coefficients

Integration by Parts identities

• An integral F can be rewritten in terms of simpler ones using
Integration by Parts (IBP) identities:

∂

∂pi
pj ◦ F = 0

=

∫
dDp1dDp2

1
(p21)n1(p22)n2(p1 − p2)2n3(Q+ p1)2n4

to which we apply the IBP identity Qµ ∂
∂pµ2

:[
−2n2

Q · p2
p22

+ 2n3
Q · p2
p23

− n3
p24
p23

+ n3
p21
p23

+ n3
Q2

p23

]
◦ F = 0

Integration by Parts identities

[
−2n2

Q · p2
p22

+ 2n3
Q · p2
p23

− n3
p24
p23

+ n3
p21
p23

+ n3
Q2

p23

]
◦ F = 0

Express IBP identity in basis {p21 ,p22,p23,p24, 2Q · p2} with raising
operators:[

−n2N+
5 N+

2 + n3
(
N+
5 N+

3 − N−
4 N+

3 + N−
1 N+

3 + Q2N+
3
)]
F = 0

• If we can combine rules such that every term has a N−
i , we can

repeat the rule and remove a propagator

Example: the triangle rule

∫
dDk kµ1 · · · kµN[

(k+ p1)2 +m2
1
]a1[(k+ p2)2 +m2

2
]a2(k2)b(p21 +m2

1)
c1(p22 +m2

2)
c2

Apply ∂
∂kµ kµ ◦ F = 0:

1 = 1
D+ N− a1 − a2 − 2b

[
a1A+

1 (B− − C−1 )+

a2A+
2 (B− − C−2 )

]

6 / 45



Example: the triangle rule

∫
dDk kµ1 · · · kµN[

(k+ p1)2 +m2
1
]a1[(k+ p2)2 +m2

2
]a2(k2)b(p21 +m2

1)
c1(p22 +m2

2)
c2

Apply ∂
∂kµ kµ ◦ F = 0:

1 = 1
D+ N− a1 − a2 − 2b

[
a1A+

1 (B− − C−1 )+

a2A+
2 (B− − C−2 )

]

6 / 45



Rules are not always easy

id Z(n1?pos_,n2?pos_,n3?pos_,n4?pos_,n5?pos_,n6?pos_,n7?pos_,
n8?pos_,n9?pos_,n10?neg0_,n11?neg0_,n12?neg0_,n13?neg0_,n14?neg_)

= -rat(1,-2*ep-2*n1-n3-n6-n12-n14+4)*(
+Z(-1+n1,-1+n2,n3,n4,1+n5,n6,n7,n8,n9,n10,n11,n12,n13,1+n14)*rat(-n5,1)
+Z(-1+n1,1+n2,n3,n4,-1+n5,n6,n7,n8,n9,n10,n11,n12,n13,1+n14)*rat(n2,1)
+Z(-1+n1,1+n2,n3,n4,n5,n6,n7,-1+n8,n9,n10,n11,n12,n13,1+n14)*rat(-n2,1)
+Z(-1+n1,n2,1+n3,n4,n5,n6,n7,n8,-1+n9,n10,n11,n12,n13,1+n14)*rat(-n3,1)
+Z(-1+n1,n2,n3,n4,-1+n5,n6,n7,n8,n9,n10,n11,1+n12,n13,1+n14)*rat(-n12,1)
+Z(-1+n1,n2,n3,n4,1+n5,n6,n7,-1+n8,n9,n10,n11,n12,n13,1+n14)*rat(n5,1)
+Z(-1+n1,n2,n3,n4,n5,n6,-1+n7,n8,n9,n10,n11,n12,1+n13,1+n14)*rat(-n13,1)
+Z(-1+n1,n2,n3,n4,n5,n6,n7,-1+n8,n9,n10,n11,1+n12,n13,1+n14)*rat(2*n12,1)
+Z(-1+n1,n2,n3,n4,n5,n6,n7,-1+n8,n9,n10,n11,n12,1+n13,1+n14)*rat(n13,1)
+Z(-1+n1,n2,n3,n4,n5,n6,n7,-1+n8,n9,n10,n11,n12,n13,2+n14)*rat(2*n14+2,1)
+Z(-1+n1,n2,n3,n4,n5,n6,n7,n8,-1+n9,n10,n11,n12,1+n13,1+n14)*rat(n13,1)
+Z(-1+n1,n2,n3,n4,n5,n6,n7,n8,n9,1+n10,-1+n11,n12,n13,1+n14)*rat(-n10,1)
+Z(-1+n1,n2,n3,n4,n5,n6,n7,n8,n9,n10,n11,1+n12,n13,n14)*rat(-n12,1)
+Z(-1+n1,n2,n3,n4,n5,n6,n7,n8,n9,n10,n11,n12,n13,1+n14)*rat(-n2+n5,1)
+Z(n1,-1+n2,-1+n3,n4,n5,n6,n7,n8,n9,n10,n11,1+n12,n13,1+n14)*rat(n12,1)
+Z(n1,-1+n2,-1+n3,n4,n5,n6,n7,n8,n9,n10,n11,n12,n13,2+n14)*rat(1+n14,1)
+Z(n1,-1+n2,1+n3,n4,n5,n6,n7,n8,-1+n9,n10,n11,n12,n13,1+n14)*rat(n3,1)

+Z(n1,-1+n2,n3,-1+n4,n5,n6,n7,n8,n9,n10,1+n11,n12,n13,1+n14)*rat(n11,1)
+Z(n1,-1+n2,n3,n4,-1+n5,n6,n7,n8,n9,n10,n11,1+n12,n13,1+n14)*rat(n12,1)
+Z(n1,-1+n2,n3,n4,n5,-1+n6,n7,n8,n9,n10,n11,n12,n13,2+n14)*rat(-n14-1,1)
+Z(n1,-1+n2,n3,n4,n5,n6,-1+n7,1+n8,n9,n10,n11,n12,n13,1+n14)*rat(2*n8,1)
+Z(n1,-1+n2,n3,n4,n5,n6,-1+n7,n8,n9,n10,1+n11,n12,n13,1+n14)*rat(-n11,1)
+Z(n1,-1+n2,n3,n4,n5,n6,-1+n7,n8,n9,n10,n11,n12,1+n13,1+n14)*rat(-n13,1)
+Z(n1,-1+n2,n3,n4,n5,n6,n7,-1+n8,n9,n10,n11,1+n12,n13,1+n14)*rat(-2*n12,1)
+Z(n1,-1+n2,n3,n4,n5,n6,n7,-1+n8,n9,n10,n11,n12,1+n13,1+n14)*rat(n13,1)
+Z(n1,-1+n2,n3,n4,n5,n6,n7,-1+n8,n9,n10,n11,n12,n13,2+n14)*rat(-2*n14-2,1)
+Z(n1,-1+n2,n3,n4,n5,n6,n7,1+n8,-1+n9,n10,n11,n12,n13,1+n14)*rat(-n8,1)
+Z(n1,-1+n2,n3,n4,n5,n6,n7,n8,-1+n9,n10,n11,1+n12,n13,1+n14)*rat(-2*n12,1)
+Z(n1,-1+n2,n3,n4,n5,n6,n7,n8,n9,1+n10,-1+n11,n12,n13,1+n14)*rat(n10,1)
+Z(n1,-1+n2,n3,n4,n5,n6,n7,n8,n9,1+n10,n11,n12,n13,1+n14)*rat(-2*n10,1)
+Z(n1,-1+n2,n3,n4,n5,n6,n7,n8,n9,n10,1+n11,n12,n13,1+n14)*rat(-n11,1)
+Z(n1,-1+n2,n3,n4,n5,n6,n7,n8,n9,n10,n11,1+n12,n13,n14)*rat(-n12,1)
+Z(n1,1+n2,n3,n4,-1+n5,n6,n7,n8,n9,-1+n10,n11,n12,n13,1+n14)*rat(-n2,1)
+Z(n1,1+n2,n3,n4,-1+n5,n6,n7,n8,n9,n10,n11,n12,n13,1+n14)*rat(-n2,1)
+Z(n1,1+n2,n3,n4,n5,n6,n7,-1+n8,n9,-1+n10,n11,n12,n13,1+n14)*rat(2*n2,1)
+Z(n1,1+n2,n3,n4,n5,n6,n7,-1+n8,n9,n10,n11,n12,n13,1+n14)*rat(n2,1)
+Z(n1,n2,-1+n3,n4,n5,n6,n7,-1+n8,n9,n10,n11,1+n12,n13,1+n14)*rat(-n12,1)
+Z(n1,n2,-1+n3,n4,n5,n6,n7,-1+n8,n9,n10,n11,n12,n13,2+n14)*rat(-n14-1,1)
+Z(n1,n2,1+n3,n4,n5,n6,n7,n8,-1+n9,-1+n10,n11,n12,n13,1+n14)*rat(n3,1)

+Z(n1,n2,n3,n4,-1+n5,n6,-1+n7,n8,n9,n10,n11,n12,1+n13,1+n14)*rat(n13,1)
+Z(n1,n2,n3,n4,-1+n5,n6,n7,n8,-1+n9,n10,n11,n12,1+n13,1+n14)*rat(-n13,1)
+Z(n1,n2,n3,n4,-1+n5,n6,n7,n8,-1+n9,n10,n11,n12,n13,2+n14)*rat(1+n14,1)
+Z(n1,n2,n3,n4,-1+n5,n6,n7,n8,n9,-1+n10,n11,1+n12,n13,1+n14)*rat(n12,1)
+Z(n1,n2,n3,n4,-1+n5,n6,n7,n8,n9,n10,-1+n11,n12,1+n13,1+n14)*rat(n13,1)
+Z(n1,n2,n3,n4,-1+n5,n6,n7,n8,n9,n10,n11,1+n12,n13,1+n14)*rat(n12,1)
+Z(n1,n2,n3,n4,-1+n5,n6,n7,n8,n9,n10,n11,n12,1+n13,1+n14)*rat(n13,1)
+Z(n1,n2,n3,n4,-1+n5,n6,n7,n8,n9,n10,n11,n12,n13,1+n14)*rat(-n2+n8-n13,1)
+Z(n1,n2,n3,n4,n5,-1+n6,n7,-1+n8,n9,n10,n11,n12,n13,2+n14)*rat(1+n14,1)
+Z(n1,n2,n3,n4,n5,n6,-1+n7,-1+n8,n9,n10,1+n11,n12,n13,1+n14)*rat(n11,1)
+Z(n1,n2,n3,n4,n5,n6,-1+n7,-1+n8,n9,n10,n11,n12,1+n13,1+n14)*rat(-2*n13,1)
+Z(n1,n2,n3,n4,n5,n6,-1+n7,n8,n9,-1+n10,1+n11,n12,n13,1+n14)*rat(n11,1)
+Z(n1,n2,n3,n4,n5,n6,-1+n7,n8,n9,n10,n11,-1+n12,n13,2+n14)*rat(1+n14,1)
+Z(n1,n2,n3,n4,n5,n6,-1+n7,n8,n9,n10,n11,n12,1+n13,1+n14)*rat(n13,1)
+Z(n1,n2,n3,n4,n5,n6,-1+n7,n8,n9,n10,n11,n12,n13,1+n14)*rat(-2*ep-2*n4-1,1)
+Z(n1,n2,n3,n4,n5,n6,1+n7,-1+n8,n9,n10,n11,n12,n13,1+n14)*rat(-n7,1)
+Z(n1,n2,n3,n4,n5,n6,n7,-1+n8,-1+n9,n10,n11,n12,1+n13,1+n14)*rat(n13,1)
+Z(n1,n2,n3,n4,n5,n6,n7,-1+n8,-1+n9,n10,n11,n12,n13,2+n14)*rat(-2*n14-2,1)
+Z(n1,n2,n3,n4,n5,n6,n7,-1+n8,n9,-1+n10,1+n11,n12,n13,1+n14)*rat(-n11,1)
+Z(n1,n2,n3,n4,n5,n6,n7,-1+n8,n9,-1+n10,n11,1+n12,n13,1+n14)*rat(-2*n12,1)
+Z(n1,n2,n3,n4,n5,n6,n7,-1+n8,n9,-1+n10,n11,n12,n13,2+n14)*rat(-2*n14-2,1)
+Z(n1,n2,n3,n4,n5,n6,n7,-1+n8,n9,1+n10,n11,n12,n13,1+n14)*rat(2*n10,1)

+Z(n1,n2,n3,n4,n5,n6,n7,-1+n8,n9,n10,-1+n11,n12,1+n13,1+n14)*rat(-2*n13,1)
+Z(n1,n2,n3,n4,n5,n6,n7,-1+n8,n9,n10,1+n11,n12,n13,1+n14)*rat(n11,1)
+Z(n1,n2,n3,n4,n5,n6,n7,-1+n8,n9,n10,n11,-1+n12,n13,2+n14)*rat(-n14-1,1)
+Z(n1,n2,n3,n4,n5,n6,n7,-1+n8,n9,n10,n11,1+n12,n13,1+n14)*rat(-2*n12,1)
+Z(n1,n2,n3,n4,n5,n6,n7,-1+n8,n9,n10,n11,n12,1+n13,1+n14)*rat(-3*n13,1)
+Z(n1,n2,n3,n4,n5,n6,n7,-1+n8,n9,n10,n11,n12,n13,1+n14)*rat(10*ep+2*n1,1)
+Z(n1,n2,n3,n4,n5,n6,n7,-1+n8,n9,n10,n11,n12,n13,2+n14)*rat(-2*n14-2,1)
+Z(n1,n2,n3,n4,n5,n6,n7,n8,-1+n9,-1+n10,n11,1+n12,n13,1+n14)*rat(-n12,1)
+Z(n1,n2,n3,n4,n5,n6,n7,n8,-1+n9,-1+n10,n11,n12,1+n13,1+n14)*rat(-n13,1)
+Z(n1,n2,n3,n4,n5,n6,n7,n8,-1+n9,n10,-1+n11,n12,n13,2+n14)*rat(-n14-1,1)
+Z(n1,n2,n3,n4,n5,n6,n7,n8,-1+n9,n10,n11,n12,1+n13,1+n14)*rat(-n13,1)
+Z(n1,n2,n3,n4,n5,n6,n7,n8,-1+n9,n10,n11,n12,n13,1+n14)*rat(-4*ep-2*n1-n3,1)
+Z(n1,n2,n3,n4,n5,n6,n7,n8,n9,-1+n10,n11,n12,n13,1+n14)*rat(-n5+1,1)
+Z(n1,n2,n3,n4,n5,n6,n7,n8,n9,1+n10,-1+n11,n12,n13,1+n14)*rat(n10,1)
+Z(n1,n2,n3,n4,n5,n6,n7,n8,n9,n10,-1+n11,n12,n13,1+n14)*rat(2*ep+n5+2*n8

+n9+n10+n11-5,1)
+Z(n1,n2,n3,n4,n5,n6,n7,n8,n9,n10,n11,1+n12,n13,n14)*rat(n12,1)
+Z(n1,n2,n3,n4,n5,n6,n7,n8,n9,n10,n11,n12,-1+n13,1+n14)*rat(-2*ep-n5-2*n8

-n9-n11-n14+3,1)
+Z(n1,n2,n3,n4,n5,n6,n7,n8,n9,n10,n11,n12,n13,1+n14)*rat(2*ep+n2+n7+

2*n8+n9+n11+n14-4,1)
);

Computational bottlenecks

• Often IBPs cannot be solved parametrically
• Solve the system through a brute force Gaussian elimination
• Simplification of coefficients with many masses and ϵ is a
bottleneck

• Millions of terms that do not fit in memory
• Swapping kills performance

Requires special tools
Mathematica, Maple, etc. cannot process this workload.

Polynomial GCDs

gcd(a,b) =
{
a, b = 0
gcd(b,a%b) otherwise

a = x8 + x6 − 3x4 − 3x3 + 8x2 + 2x− 5
b = 3x6 + 5x4 − 4x2 − 9x+ 21

{
a, b = 0
gcd(b,a%b) otherwise

a = x8 + x6 − 3x4 − 3x3 + 8x2 + 2x− 5
b = 3x6 + 5x4 − 4x2 − 9x+ 21

R1 = −5
9x

2 +
1
9x

2 − 1
3

R2 = −117
25 x

2 − 9x+ 411
25

R3 = −102500
6591 +

233150
19773 x

R4 = −1288744821
543589225

• From Z to Q

• Rapid coefficient growth

Polynomial GCDs

An attempt with pseudo-remainders:

R1 = −15x4 + 3x2 − 9
R2 = 15795x2 + 30375x− 5953
R3 = 1254542875143750x− 1654608338437500
R4 = 12593338795500743100931141992187500

• Coefficients still in Z, but are huge
• Modular algorithm with reconstruction is needed
• Multivariate case is much harder and actively researched

FORM

• FORM [Vermaseren '89] is a very popular symbolic manipulation
toolkit in the field

• It processes each term one by one, as not to run out of memory
• Processed terms are stored into memory at first
• When memory is full, write and read from disk
• Can handle large scale computations

Obtaining FORM

• Download FORM 4.3.1 from:
https://github.com/vermaseren/form/releases

• Tutorial: https://www.nikhef.nl/~form/maindir/
documentation/tutorial/online/online.html

• Reference manual: https://www.nikhef.nl/~form/maindir/
documentation/reference/online/online.html

• Visual Studio Code Syntax highlighting extension
• FORM Cookbook:
https://github.com/vermaseren/form/wiki/FORM-Cookbook

• Compile:
./configure
make -j4
make install

Example program

Save the following as prog1.frm:

1 Symbols a,b;
2 Local F = (a+b)^2;
3 Print;
4 .end

and run

form prog1

Example program

1 Symbols a,b; * define symbols
2 Local F = (a+b)^2; * define expression
3 Print; * print the expression
4 .end; * end the program (and sort)

Time = 0.00 sec Generated terms = 3
F Terms in output = 3

Bytes used = 108

F =
b^2 + 2*a*b + a^2;

Operations on terms

FORM can replace patterns in terms with id:

1 Symbols a,b,c;
2 Local F = (a+b)^6;
3 id a^2*b = c; * replacement
4 Print;
5 .end

F =
15*c^2 + 15*b^3*c + b^6 + 20*a*b^2*c
+ 6*a*b^5 + 6*a^3*c + a^6;

FORM behaviour

• FORM always expands terms
• FORM processes expressions term by term
• This means that only 1 term should fit in memory, the rest can
be read/written to disk

• Pro: memory is no limitation
• Con: operations on expressions are more difficult

When confused why certain operations don't exist, imagine that
every expression is too big to fit in memory

The following FORM program will run (try it):

1 Auto Symbols x; * all starting with x is a symbol
2 Local F = (x1+x2+x3+x4+x5+x6)^100;
3 .end

Time = 0.45 sec Generated terms = 100000
F 1 Terms left = 100000

Bytes used = 6106364

Time = 1.00 sec Generated terms = 200000
F 1 Terms left = 200000

Bytes used = 12519468
....

20 / 45



The following FORM program will run (try it):

1 Auto Symbols x; * all starting with x is a symbol
2 Local F = (x1+x2+x3+x4+x5+x6)^100;
3 .end

Time = 0.45 sec Generated terms = 100000
F 1 Terms left = 100000

Bytes used = 6106364

Time = 1.00 sec Generated terms = 200000
F 1 Terms left = 200000

Bytes used = 12519468
....

20 / 45



Term flow

Sorting

• At the end of a module, terms should be sorted to see if terms
will merge

• A module is ended with .sort (or .end)
• Where to place the sort is up to the user

Sorting: good vs bad

1 Symbols a,b,c,d;
2 Local F = (a+b+c+1)^6;
3 id a = -c+d+1;
4 id b = -d+1;
5 Print;
6 .end

Generates 924 terms and has 1 in the output...

Sorting: good vs bad

1 Symbols a,b,c,d;
2 Local F = (a+b+c+1)^6;
3 id a = -c+d+1;
4 .sort
5 id b = -d+1;
6 .end

First sort:

Generated terms = 462
Terms in output = 28

Second sort:

Generated terms = 84
Terms in output = 1

Identity statements I

• Patterns for id statements can be any terms
• Use wildcards var? to match any object of the same type

1 S x,y; * short for Symbol
2 L F = x^2 + y;
3 id x? = 5; * match any symbol with x?
4 Print;
5 .end

yields

F = 30

Identity statements II

Patterns can be more complicated:

1 S x,y,n;
2 L F = x^2 + y;
3 id x?^n? = x^(n + 1);
4 Print;
5 .end

yields

F = y^2 + x^3

Identity statements III

• Restrictions can be placed by a set {1,..} or a number range
{>5}

• A statement can be repeated with repeat

1 S x,n;
2 L F = x^10;
3 repeat id x^n?{>1} = x^(n-1) + x^(n-2);
4 Print;
5 .end

yields

F = 34 + 55*x;

Functions

• Functions for non-commutative functions
• CFunctions for commutative functions

1 S a,b,c;
2 CF f; * short for CFunctions
3 Local F = f(1,2,c);
4 id f(1,2,b?) = f(1,2,b?+1);
5 Print;
6 .end

yields:

F = f(1,2,c+1);

Symmetric functions

1 CF f(s);
2 L F = f(3,2,1);
3 Print;
4 .end

yields:

F = f(1,2,3);

Ranged wildcards

A wildcard starting with a ? indicates a range:

1 S x;
2 L F = f(1,2,x,3,4);
3 id f(?a,x,?b) = f(?b,?a);
4 Print;
5 .end

yields

F = f(3,4,1,2);

Applying statements to arguments

id-statements are only applied at ground-level:

1 S x,y;
2 L F = f(x*y);
3 id x = 5; * does not match
4 argument f;
5 id x = 6;
6 endargument;
7 Print;
8 .end

yields

F = f(6*y)

If statements

1 S x,y;
2 L F = f(2) + f(5);
3 if (match(f(x?{>4})));
4 id f(x?) = f(x + 1);
5 else;
6 id f(x?) = f(x - 1);
7 endif;
8 Print;
9 .end

yields

F = f(1) + f(6)

Bracketing I

• Powers of variables can be extracted
• The terms are not nested for real, but information about
brackets can be used in the next module

1 S x,y,z;
2 L F = x*y + x^2*y + x^2*z + 2;
3 Bracket x; * extract powers of x
4 Print;
5 .end

F = + x * ( y )
+ x^2 * ( z + y )
+ 2;

Bracketing II

• Brackets can be indexed in the next module

1 S x,y,z;
2 L F = x*y + x^2*y + x^2*z + 2;
3 Bracket x;
4 .sort
5 L G = F[x^2];
6 Print G; * only print G
7 .end

G = z + y

Bracketing III

• Bracketed content can be collected in a function if it fits in
memory

1 S x,y,z;
2 L F = x*y + x^2*y + x^2*z + 2;
3 Bracket x;
4 .sort
5 CF f;
6 Collect f;
7 Print;
8 .end

F = f(z + y)*x^2 + f(y)*x + f(2)

Tools for physicists



Vectors and indices

Contraction and Einstein summation:

1 Index i1,i2,i3;
2 Vector p1,p2,p3;
3 Local F = p1(i1)*(p2(i1)+p3(i3))*(p1(i2)+p2(i3));
4 Print;
5 .end

yields:

F = p1(i1)*p1(i2)*p3(i3) + p1(i1)*p2.p3
+ p1(i2)*p1.p2 + p2(i3)*p1.p2;

Make sure an index does not appear more than twice in a term!

Traces and gamma matrices

1 S D;
2 Index i1=D,i2=D; * D-dimensional indices
3 Vector p1,p2;
4 Local F1 = g_(1, i1, i1); * gamma matrices
5 Local F2 = g_(1, p1, i2);
6 Local F3 = g_(1, p1, p2);
7 tracen 1; * n-dimensional trace of spin line 1
8 Print;
9 .end

F1 = 4*D;
F2 = 4*p1(i2);
F3 = 4*p1.p2;

Feynman rule application

1 S vhhg, gh, gl; * ghost-gluon vertex, ghost, gluon
2 I i1,i2;
3 V Q,p1,p2,p3,p4;
4 CF vx,prop;
5 L F = vx(Q,p1,p2,i1,vhhg)
6 *vx(-p1,-Q,-p2,i2,vhhg)
7 *prop(p1,i1,i2,gl)*prop(p2,gh);
8

9 id prop(p1?,i1?,i2?,gl) = d_(i1,i2)/p1.p1;
10 id prop(p1?,gh) = 1/p1.p1;
11 id vx(p1?,p2?,p3?,i1?,vhhg) = -i_*vx(p1,p2,p3)*p1(i1);

F = vx(-p1,-Q,-p2)*vx(Q,p1,p2)*Q.p1*p1.p1^-1*p2.p2^-1;

Rational polynomials

Create rational polynomials using polyratfun:

1 S x, y;
2 CF rat;
3 polyratfun rat; * enable polyratfun
4

5 L F = rat(y,x) + rat(x,1)*rat(1,y+1);
6 Print;
7 .end

F = rat(x^2 + y^2 + y,x*y + x)

IBP reduction of one-loop massive vacuum bubble

1 * rewrite k.k in the numerator
2 repeat id k1?.k1?*prop(k1?,n1?) =
3 prop(k1, n1-1) + m^2 * prop(k1, n1);
4 id prop(k1?,n?) = prop(n);
5

6 * 1-loop IBP
7 id prop(n1?{<1}) = 0;
8 repeat id prop(n1?{>1}) = prop(-1 + n1)*
9 rat((2 + (4-2*ep) - 2*n1), 2* (-1 + n1)) / m^2;
10 * master integral expanded in ep
11 id prop(1) = 1/ep + 1;

40 / 45



Exercises I

1. Get the maximum function argument, e.g. f(1,4,2,5,3) → f(5)
2. Represent a graph using vertex functions, e.g.

vx(Q,p1,p2)*vx(-Q,-p1,-p2) and:
2.1 Write an algorithm that checks if a graph is connected by shrinking

edges
2.2 Determine the number of loops, using L = E− V+ 1

Large exercise preparation

• Obtain Qgraf from
http://cfif.ist.utl.pt/~paulo/qgraf.html

• Download FORM configuration from https://gist.github.com/
benruijl/c37a8029a2b47a618dd1d2d46c631249

• Compile with gfortran

• Generate ϕ3 and QED input file

Large exercise

• Apply Feynman rules for two-loop photon self-energy graphs in
QED

• Do the same for two-loop photon self-energy with QCD
corrections

• Apply triangle reduction rule
• Apply one-loop master formula (next slide)

One-loop reduction

∫ dDk
(2π)D

Pn(k)
k2α(k− Q)2β =

1
(4π)2 (Q

2)D/2−α−β

[n/2]∑
σ≥0

G(α, β,n, σ)Q2σ
{

1
σ!

(
□
4

)σ

Pn(k)
}

k=Q

where
Pn(k) = kµ1kµ2 · · · kµn , □ = ∂2/∂kµ∂kµ ,

G(α, β,n, σ) = (4π)ϵ Γ(α+ β − σ − D/2)Γ(D/2− α+ n− σ)Γ(D/2− β + σ)

Γ(α)Γ(β)Γ(D− α− β + n)

• Implement d'Alambertian using distrib_ and dd_

Hints

• Use sets to map numerical indices to Lorentz/Dirac indices
• Represent dirac algebra as gamma(i,p,j) and chain the functions

Good luck!

